

FORWARD MODELING COMPUTER PROGRAM FOR THE VERY LOW FREQUENCY,
RADIO-WAVE, TERRAIN-RESISTIVITY ELECTROMAGNETIC METHOD: VLF.BAS

by Deborah G. Grantham, F. P. Haeni and David L. Mazzaferro

UNITED STATES GEOLOGICAL SURVEY

Open-File Report 86-407W

Prepared in cooperation with
DEPARTMENT OF ENERGY

Hartford, Connecticut

1986

U.S. DEPARTMENT OF THE INTERIOR

DONALD PAUL HODEL, Secretary

GEOLOGICAL SURVEY

Dallas L. Peck, Director

For additional information,
write to:
Chief, Connecticut Office
U.S. Geological Survey, WRD
450 Main Street, Room 525
Hartford, Connecticut 06103

Copies of this report can be
purchased from:
Books & Open File Reports Section
Western Distribution Branch
U.S. Geological Survey
Box 25425, Federal Center
Denver, CO 80225
Telephone: (303) 236-7476

CONTENTS

	Page
Abstract.....	1
Introduction.....	2
The VLF.BAS program.....	2
Acknowledgments.....	2
General principles of a very low frequency resistivity instrument.....	4
Program calculations.....	5
Assumptions.....	5
VLF.BAS algorithm.....	5
User procedures.....	10
Loading the program.....	10
Adapting to other computer systems.....	10
Data input.....	10
Interacting with the program.....	10
Output.....	11
Example problem.....	11
Selected references.....	19
Glossary.....	21
Attachment A: Listing of the code for the VLF.BAS program.....	24

ILLUSTRATIONS

	Page
Figure 1. Diagram of induced electric and magnetic fields....	3
2. Photograph of commercial very low frequency instrument.....	4
3. Diagram of layered Earth model parameters and notation scheme.....	8

TABLES

Table 1. Input data for example problem.....	12
2. VLF.BAS output data and format.....	15
3. Hypothetical very low frequency field survey data.....	16
4. VLF.BAS output data and format for interpreted Earth model.....	17

CONVERSION FACTORS

The following report uses metric (International System) units as the primary system of measurement. The units commonly are abbreviated using the notations shown in parentheses. Metric units can be converted to inch-pound units by multiplying by the factors given in the following list.

<u>Multiply Metric Units</u>	<u>by</u>	<u>To Obtain Inch-Pound Units</u>
meter (m)	3.2810	foot (ft)
millisiemens per meter (mS/m)	0.3048	millimhos per foot (mmho/ft)

FORWARD MODELING COMPUTER PROGRAM FOR THE VERY LOW FREQUENCY,
RADIO-WAVE, TERRAIN-RESISTIVITY ELECTROMAGNETIC METHOD: VLF.BAS

by Deborah G. Grantham, F. P. Haeni and David L. Mazzaferro

ABSTRACT

Program VLF.BAS is a BASIC computer program that calculates the apparent resistivity and phase angle that would be measured by very low frequency terrain-resistivity instruments at the surface of the Earth for a given sequence of conductive or resistive horizontal layers. The program can be used to determine the feasibility of using the very low frequency technique in a field investigation and to interpret the field results. The resistivity and thickness of individual subsurface layers can be iteratively adjusted to minimize the difference between the calculated and observed apparent resistivity and phase angle. The program is interactive and has monitor display or hardcopy output. VLF.BAS is written for a microcomputer but is also running on the U.S. Geological Survey's Distributed Information System computer system.

INTRODUCTION

The VLF.BAS Program and the Very Low Frequency Technique

This documentation describes the use of a BASIC ^{1/} computer program designed to calculate the apparent resistivity and phase angle as measured by VLF (very low frequency) terrain-resistivity instruments at the surface of a layered Earth. The program is useful for planning field investigations and interpreting field observations.

A VLF radio signal induces horizontal electric and magnetic fields in the Earth that are related to the electrical properties of the Earth (see fig. 1). The VLF.BAS program utilizes the relationships between these fields and the electrical properties of the Earth to calculate the apparent resistivity and the phase angle at the surface. The apparent resistivity at the surface is a function of the frequency of the signal used in the measurements and the relationship between the induced horizontal electric and magnetic fields.

The phase angle--the angle between the primary horizontal magnetic field and the horizontal electric field--is a function of the spatial ordering of the resistivity structure of the Earth. A resistive above a conductive layer produces a phase angle greater than 45 degrees, a conductive over a resistive layer produces a phase angle of less than 45 degrees, while a homogeneous Earth produces a phase angle of 45 degrees (Geonics Ltd., 1979).


Acknowledgements

Acknowledgment is due Frank Frischknecht, U.S. Geological Survey, Denver, Colorado, for providing the authors the algorithm for VLF.BAS, and to Duncan McNeill, Geonics Ltd., Ontario, Canada, for technical information and review.

1/

The use of brand, trade, or firm names in this report is for identification purposes only and does not constitute endorsement by the U.S. Geological Survey.

**REMOTE VLF
TRANSMITTER**



$$\text{Apparent terrain resistivity } (\rho_a) = \frac{0.2}{\text{frequency}} \left(\frac{E_x}{H_y} \right)^2$$

and, the exploration depth (δ) = $500 \sqrt{\rho / \text{frequency}}$

where: E_z = vertical electrical field

E_x = horizontal radial electrical field

H_y = horizontal radial magnetic field

$\rho_1 \ll \rho_2$ = relative resistivity of conductive and resistive ground

(Modified from Collett, 1978, p.30).

Figure 1: Electric and magnetic fields induced in the Earth by a plane-wave radio signal.

GENERAL PRINCIPLES OF A VERY LOW FREQUENCY RESISTIVITY INSTRUMENT

The EM16R (see fig. 2) is one of several commercially available instruments that can be used to conduct a VLF resistivity survey. The equipment measures the ratio of the horizontal components of the electric and magnetic fields and is electronically calibrated using equations (1) and (3) so that the apparent resistivity and the phase angle can be read directly from the instrument. The VLF signal originates at stations throughout the world and operated by both the United States and other countries. The VLF stations operate at a narrow band of frequencies in the range of 3 to 30 kHz.



Figure 2: One commercially available model of a Very Low Frequency system.

PROGRAM CALCULATIONS

Assumptions

In the application of electromagnetic wave theory to geophysical investigations, several simplifying assumptions are made:

- 1) VLF radio frequencies (3 to 30 kHz) are used and ground resistivity is assumed to be less than 10,000 ohm-m so that displacement currents can be assumed to be negligible (Duncan McNeill, Geonics Limited, written commun., 1985).
- 2) Signals are from distant transmitters such that the study areas are in the transmitter's far field zone (several wavelengths of the transmitted wave from the transmitter) and the primary field components (E_z and H_y) are plane polarized (see fig. 1).
- 3) Intermittent sources of electrical interference, such as atmosphere disturbances, are neglected; the investigator must also be aware of man-made sources of electrical interferences, such as fences, buried pipes or power lines, that may give rise to anomalous readings because they carry VLF currents (Keller and Frischknecht, 1966).
- 4) The Earth's subsurface consists of laterally homogeneous, infinite layers (Telford and others, 1976).

VLF.BAS Algorithm

The VLF.BAS program calculates the apparent resistivity that would be measured at the surface of the Earth using:

$$\rho_a = (0.2/f) \left| \frac{E_x}{H_y} \right|^2 \quad (1)$$

in which ρ_a = apparent resistivity
(ohm-m),
 f = signal frequency of
the VLF transmitter
(kHz), and
 $\left| \frac{E_x}{H_y} \right|$ = magnitude of the ratio
of the horizontal
electric (E_x) to the
horizontal magnetic
(H_y) fields (ohms).

Equation (1) can also be written in terms of the wave impedance, Z:

$$\rho_a = (0.2/f) |Z|^2 \quad (2)$$

in which ρ_a = apparent resistivity
 (ohm-m).
 f = signal frequency of the VLF transmitter (kHz), and
 $Z = E/H$ (a complex number having a real and an imaginary part) (ohms).

The phase angle is calculated using:

$$\Phi = \arctan [I(Z)/R(Z)] \quad (3)$$

in which Φ = phase angle (radians),
 $Z = E/H$ (ohms),
 $I(Z)$ = imaginary part of Z (ohms), and
 $R(Z)$ = real part of Z (ohms).

The phase angle is converted to degrees in the program.

Both the apparent resistivity and the phase angle (Equations 1 and 3) are calculated from the wave impedance at the surface, Z. The impedance of a homogeneous conductor is the intrinsic impedance, defined by:

$$\eta = E/H = \left(\begin{array}{c} i\mu\omega \\ 0 \\ \sigma \end{array} \right)^{1/2} \quad (\text{ohms}) \quad (4)$$

in which η = intrinsic impedance (ohms),
 i = square root of -1,
 μ_0 = magnetic permeability of free space (henrys/m),
 ω = angular frequency of the signal = $2\pi f$ (kHz), and
 σ = conductivity (mS/m).

For a homogeneous Earth, but not for a layered Earth, the intrinsic impedance and the wave impedance are identical. To calculate the wave impedance at the surface of a layered Earth, the additional concept of skin depth is needed.

Skin depth is the depth of penetration of a wave passing into a conductor in which the amplitude of the wave is attenuated to 1/e of its amplitude at the surface of the conductor. Skin depth is defined by:

$$\delta = \left(\frac{2}{\mu_0 \sigma \omega} \right)^{1/2} = 500 (\rho / f)^{1/2} \quad (5)$$

- in which δ = skin depth (m),
- μ_0 = magnetic permeability
of free space = $4\pi \times 10^{-7}$ (henrys/m),
- ω = angular frequency (kHz),
- ρ = resistivity of the conductor (ohm-m), and
- f = signal frequency (kHz).




Figure 3: Layered Earth-model parameters and notation.

The wave impedance of a layered Earth is calculated by starting with the deepest layer, which has an intrinsic impedance calculated from equation (4) and a skin depth calculated from equation (5), and calculating upwards through the shallowest layer using the following algorithm (Duncan McNeill, Geonics Ltd., written commun., 1985):

$$Z_{n-1} = \eta_{n-1} \frac{\eta_n + \eta_{n-1} \tanh[(1+i)h_{n-1} / \delta_{n-1}]}{\eta_{n-1} + \eta_n \tanh[(1+i)h_{n-1} / \delta_{n-1}]} \quad (6)$$

in which $n, n-1$ = indices referring to the n th and $(n-1)$ th layers (see fig. 3),

η_{n-1} = intrinsic impedance (ohms),

δ_{n-1} = skin depth (m),

h_{n-1} = thickness (m),

Z_{n-1} = wave impedance (ohms), and
 i = square root of -1 .

$$Z_j = \eta_j \frac{Z_{j+1} + \eta_j \tanh[(1+i)h_j / \delta_j]}{\eta_j + Z_{j+1} \tanh[(1+i)h_j / \delta_j]} \quad (7)$$

$$Z_2 = \eta_2 \frac{Z_3 + \eta_2 \tanh[(1+i)h_2 / \delta_2]}{\eta_2 + Z_3 \tanh[(1+i)h_2 / \delta_2]} \quad (8)$$

$$Z_1 = \eta_1 \frac{Z_2 + \eta_1 \tanh[(1+i)h_1 / \delta_1]}{\eta_1 + Z_2 \tanh[(1+i)h_1 / \delta_1]} \quad (9)$$

Equation (9) gives the wave impedance used to calculate the apparent resistivity and the phase angle, using equations (1) and (3). Equations (6) through (9) are derived in detail in Wait (1982, p.155).

USER PROCEDURES

Loading The Program

An IBM or compatible personal computer with one floppy disk drive and a minimum of 64K bytes of memory is required to run this program. The program runs significantly faster on a system that has an 8087 math coprocessor installed.

Adapting to Other Computer Systems

The VLF.BAS program was written for an IBM-PC but is adaptable to other computer systems. The program is currently running on the U.S. Geological Survey's DIS (Distributed Information System) computer system with little modification from the IBM personal computer version.

Data Input

The program interactively requests a site name, number of layers assumed to be present, transmitting frequencies (in kHz), the resistivities (in ohm-m) and the thicknesses (in m) of the subsurface layers. Through these parameters the user defines a model. Layer resistivity and thickness values can be based on direct-current electrical, seismic-refraction or borehole-geophysical surveys; geologic knowledge; test holes; wells; or other sources.

Interacting with the Program

As the program runs, it will prompt the user for the input data. All of the responses should be in upper case letters and entered by striking the RETURN key.

There is opportunity, after the initial data input and after each computational run, to change all or some of the input parameters. Typically, parameter changes would be made when trying to match the model program with field data. Changes should be made after the relationships between the layer resistivities and spatial ordering of the conductive and resistive layers have been considered. For example, if, in a two-layer problem, an increase in apparent resistivity is needed with no change in the phase angle, then a similar increase in the resistivity of both layers should produce the desired result. On

the other hand, if an increase in resistivity and a decrease in the phase angle is needed, only the resistivity of the lower layer should be increased. This conceptual planning will be more productive and efficient than random changes in the data.

Upon completing a problem, the user can change any of the input parameters, can begin a new problem or can terminate the program and return to BASICA by responding appropriately to the prompts.

Output

The program outputs to the printer as well as to the monitor. The output consists of a listing of the input values of layer thicknesses and resistivities and the calculated values of apparent resistivity and phase angle for each VLF frequency.

Example Problem

A typical study area where the VLF technique could be used is a landfill site from which it is suspected that conductive contaminants are being leached into the groundwater. The first problem is to decide whether or not the VLF technique would detect the conductor contaminant plume at this site. This can be accomplished by testing a preliminary subsurface model with the VLF.BAS program, based on existing hydrogeologic information. From borehole and geologic data, it is known that the subsurface in uncontaminated areas can be generalized as three layers: a resistive layer representing the unsaturated material (2000 ohm-m and 10 m thick) overlying a conductive layer representing the saturated material (400 ohm-m and 40 m thick) overlying resistive bedrock (800 ohm-m).

In the contaminated areas of the landfill, the subsurface can be generalized as three layers: a resistive, unsaturated layer (2000 ohm-m and 10 m thick) overlying a saturated, contaminated zone (50 ohm-m and 40 m thick) overlying resistive bedrock (800 ohm-m).

Two VLF signals commonly used in the northeast are transmitted from Annapolis, Maryland, (21.4 kHz) and Cutler, Maine, (24.0 kHz as of 1985; 17.8 kHz prior to 1985).

The input data needed for the sample run of VLF.BAS are listed in table 1.

Table 1.--Input data for example problem

	<u>Uncontaminated area</u>	<u>Contaminated area</u>
Number of layers	3	3
Number of frequencies	2	2
Layer 1		
Resistivity (ohm-m)	2000	2000
Thickness (m)	10	10
Layer 2		
Resistivity (ohm-m)	400	50
Thickness (m)	40	40
<u>1/</u>		
Layer 3		
Resistivity (ohm-m)	800	800
Frequency 1 (kHz)	24.0	24.0
Frequency 2 (kHz)	21.4	21.4

1/

The thickness of the deepest layer is assumed to be infinite and is arbitrarily assigned the value 10,000 meters.

The VLF.BAS program can now be used to calculate the values for apparent resistivity and phase angle that would be measured by a VLF instrument in both the uncontaminated areas and then in the contaminated area of the landfill.

To use the VLF.BAS program on a personal computer, invoke BASICA from DOS (in drive A) and load and run the program (in drive B) from BASICA by typing in the following commands:

```
BASICA (RETURN)
LOAD B:VLF.BAS (RETURN)
RUN (RETURN)
```

After an introduction of the program, the user will be prompted to continue by striking the RETURN key. The user will then be prompted to name the study site, using no more than 40 upper case letters. Respond with:

SAMPLE_SITE (RETURN)

The next prompt will be for a choice of hardcopy or monitor data display. If a hardcopy is desired, ready the printer and input:

P (RETURN)

If only a monitor display is required, strike the RETURN key.

The user will then be prompted to input the number of layers assumed for the earth model, the number of VLF stations used, the resistivities in ohm-m and thicknesses in m of each layer and the frequencies in kHz. Respond at each prompt with the value and strike the RETURN key:

ENTER # OF LAYERS
3 (RETURN)
ENTER # OF FREQUENCIES
2 (RETURN)
ENTER THE RESISTIVITY OF LAYER 1 IN OHM-METERS
2000 (RETURN)
ENTER THE THICKNESS OF LAYER 1 IN METERS
10 (RETURN)
(etc.)

The program will then display the input data and ask for corrections. If the data are correct, enter in upper case:

Y (RETURN)

If the data are incorrect, enter:

N (RETURN)

and then respond to the subsequent prompts for corrections and confirmation. Once the data have been corrected and confirmed, the calculations will be made and the frequencies and corresponding apparent resistivities and phase angles will be displayed and printed.

A prompt for continuation will appear on the monitor. Entering E will terminate the program, N will allow all new data and C will allow changes to be made in the data. Respond with:

C (RETURN)

The original data will be displayed and the user will have an opportunity to change the data. Calculations can now be made for the contaminated area of the landfill by changing the resistivity of layer two from 400 ohm-m to 50 m using the following sequence of commands:

```
ARE THE DATA CORRECT?
N (RETURN)
ENTER NUMBER OF LAYER TO BE CORRECTED
2 (RETURN)
ENTER R FOR RESISTIVITY OR T FOR THICKNESS
R ( RETURN)
ENTER NEW VALUE FOR RESISTIVITY OF LAYER 2
50 (RETURN)
ARE THE DATA CORRECT?
Y (RETURN)
```

The apparent resistivities and phase angles for the two frequencies in the contaminated area will be displayed and printed, followed by the subsurface parameters used in the calculations.

The user then terminates the program and exits from BASICA:

```
ENTER RETURN TO CONTINUE
(RETURN)
ENTER E TO EXIT PROGRAM
      C TO CORRECT PARAMETERS
      N TO ENTER NEW DATA
E (RETURN)
SYSTEM (RETURN)
```

The user will be returned to the IBM-PC operating system.

The calculations made by the VLF.BAS program, summarized in table 2, indicate that the differences in the apparent resistivities and phase angles that be would measured if a VLF survey was conducted over the uncontaminated and the contaminated areas would be large. This would make detection of the conductive contaminated groundwater plume possible.

After the preliminary model has shown that the VLF technique can detect the plume, a VLF resistivity survey of the landfill site could be planned and conducted. Field measurements of the apparent resistivity and phase angle could be made on a series of transects perpendicular to the estimated axis of the plume of contamination. The orientation of the plume would be assumed to be in the direction of groundwater flow, as determined from the

available subsurface data. The VLF field values for one hypothetical transect are summarized in table 3.

Table 2.--VLF.BAS Output data and format

UNCONTAMINATED AREA

Data Summary For Run Number 1

<u>FREQUENCY</u>	<u>RESISTIVITY</u>	<u>PHASE ANGLE</u>
2.40E+04	5.24E+02	4.61E+01
2.14E+04	5.26E+02	4.55E+01

LAYER PARAMETERS

2000
10

400
40

800
10000

CONTAMINATED AREA

Data Summary For Run Number 1

<u>FREQUENCY</u>	<u>RESISTIVITY</u>	<u>PHASE ANGLE</u>
2.40E+04	1.06E+02	6.20E+01
2.14E+04	1.00E+02	6.14E+01

LAYER PARAMETERS

2000
10

50
40

800
10000

Table 3.--Hypothetical Very Low
Frequency field survey data

<u>Station</u>	<u>Apparent resistivity (ohm-m)</u>	<u>Phase angle</u> (degrees)
1	563	46
2	570	47
3	555	47
4	560	48
5	560	49
* 6	150	53
* 7	145	56
* 8	140	57
* 9	150	57
* 10	148	59
11	560	46
12	557	48
13	570	47
14	572	45
15	561	46
16	560	49
17	555	48

* Values interpreted as contaminated groundwater.

The VLF.BAS program can now be used to generate a subsurface model whose calculated apparent resistivity and phase angle match the observed data at each field station. From the hypothetical field data in table 3, it is obvious that stations 1 through 5 and 11 through 17 represent one subsurface condition (the uncontaminated area), and stations 6 through 10 represent another condition (the contaminated area). The original model (table 1) can now be adjusted until the calculated values closely match the average field values of apparent resistivity and phase angle. This iterative modeling procedure yields an Earth model (table 4) that has a saturated layer with resistivities that are higher than the initial estimated resistivities (table 1). The phase angles show the same relationships as originally modeled--a resistive over a conductive layer.

The final interpreted Earth models, summarized in table 4, are ones that reproduce the observed values of apparent resistivity and phase angle. These models are not unique solutions for the respective field data and therefore must be kept consistent with the available geologic data.

Table 4.--Interpreted Earth model of landfill site

INTERPRETED MODEL UNCONTAMINATED

Data Summary For Run Number 1

<u>FREQUENCY</u>	<u>RESISTIVITY</u>	<u>PHASE ANGLE</u>
2.40E+04	5.61E+02	4.70E+01
2.14E+04	5.61E+02	4.64E+01

LAYER PARAMETERS

2000
10

449
48

800
10000

INTERPRETED MODEL CONTAMINATED

Data Summary For Run Number 1

<u>FREQUENCY</u>	<u>RESISTIVITY</u>	<u>PHASE ANGLE</u>
2.40E+04	1.46E+02	5.61E+01
2.14E+04	1.40E+02	5.48E+01

LAYER PARAMETERS

2000
10

91
37

800
10000

In this example, the VLF.BAS program was first used to determine the feasibility of using the VLF technique for a particular field problem and then used to interpret the field results. This is a typical procedure and results in increased efficiency of field operations and data interpretation.

Case histories of the VLF technique applied to groundwater and other geologic problems have been published by Fraser (1969), Patterson and Ronka (1971), Telford and others (1977), Duran and Haeni (1982), Greenhouse and Slaine (1982), Greenhouse and Harris (1983), and Grady and Haeni (1984). A study using the VLF.BAS program to determine the feasibility of using the VLF technique and to subsequently interpret field data has been conducted by Haeni (1986).

References Cited

- Collett, L. S., 1978, Introduction to hydrogeophysics: International Association of Hydrogeologists, Canadian Chapter, National Hydrogeological Conference and Field Trips, Edmonton, Alberta, September 30-October 4, 1978, p. 16-35.
- Duran, P. B. and Haeni, F. P., 1982, The use of electromagnetic conductivity techniques in the delineation of ground-water leachate plumes: Proceedings of the Northeast Conference on the Impact of Waste Storage and Disposal on Ground-Water Resources, Ithaca, New York, June 28-July 1, 1982, p. 8.4.1-8.4.33.
- Fraser, D. C., 1969, Contouring of VLF-EM data: Geophysics, v. 34 no. 6, p. 958-967.
- Geonics Ltd., 1979, Operating manual for EM16R, VLF resistivity meter: Geonics Ltd., Mississauga, Ontario, 37 p.
- Grady, S. L. and Haeni, F. P., 1984, Application of electromagnetic techniques on determining distribution and extent of ground-water contamination at a sanitary landfill, Farmington, Connecticut: Proceedings of Surface and Borehole Geophysical Methods in Ground-Water Investigations, San Antonio, Texas, February 7-9, 1984, p. 338-367.
- Greenhouse, J. P., and Slaine, D. D., 1982, Case studies of geophysical contaminant mapping at several waste disposal sites: Proceedings of Second National Symposium on Aquifer Restoration and Ground Water Monitoring, Columbus, Ohio, May 26-28, 1982, p. 299-315.
- Greenhouse, J. P., and Harris, R. D., 1983, Migration of contaminants in groundwater at a landfill: Case study number 7: DC, VLF and inductive resistivity surveys: Journal of Hydrology, v. 63, p. 177-197.
- Haeni, F. P., 1986, The use of combined surface geophysical techniques for determining the geometry and characteristics of sand and gravel aquifer in the glacial northeast: Proceedings of the National Water Well Association on Surface and Borehole Geophysical Methods and Ground-Water Instrumentation, October 15-17, 1986, Denver, Colorado, p.
- Keller, G. V., and Frischknecht, F. C., 1966, Electrical Methods in Geophysical Prospecting: Oxford, Pergamon Press, 517 p.

- McNeill, J. D., 1980, Electromagnetic terrain conductivity measurement at low induction numbers: Geonics, Ltd., Technical TN-6, Mississauga, Ontario, 15 p.
- Paterson, N. R. and Vaino Ronka, 1971, Five years of surveying with the VLF-EM method: Geoexploration, v. 9, p. 7-26.
- Telford, W. M., Geldart, L. P., Sheriff, R. E., and Keys, D. A., 1976, Applied Geophysics: Cambridge, Cambridge Press, 860 p.
- Telford, W. M., King, W. F., and Becker, A., 1977, VLF mapping of geologic structure: Canada Geological Survey Paper 76-25, p. 1-13.
- Wait, J. R., 1982, Geo-Electromagnetism: New York, Academic Press, 268 p.
- Zohdy, A. A. R., Eaton, G. P., and Mabey, D. R., 1974, Application of surface geophysics to ground-water investigations: U.S. Geological Survey Techniques of Water-Resources Investigations, Book 2, Chapter D1, 116 p.

GLOSSARY

Angular frequency: Rate of repetition measured in radians/second; where f = frequency in Hertz, angular frequency is $2\pi f$.

Apparent resistivity: Resistivity measured using the VLF technique that differs from true resistivity if there are inhomogeneities of the Earth; units are ohm-m.

Byte: Computer unit of binary digits usually in eight bits representing two numerals or one character.

Conductivity: Ability of material to conduct electrical current; in an isotropic material, the reciprocal of resistivity; units are siemen/m.

Disk operating system (DOS): Machine instructions and procedures for operating a computer disk drive.

Displacement currents: Currents resulting from capacitative properties of a conductor. Displacement currents in the Earth are usually negligible compared to conduction currents.

e: Base of natural logarithm; $e = 2.7183$.

i: in complex number plane, $i = \text{square root of } -1$.

Impedance: Opposition to the flow of alternating current, analogous to resistance in a direct current circuit. In inductive electromagnetic methods, the impedance is the ratio of a horizontal component of an electric field to the orthogonal horizontal component of the associated magnetic field; impedance is a complex number due to the phase differences between the electric and magnetic fields; units are ohms.

Intrinsic impedance: Impedance of a homogeneous conductor; units are ohms.

Magnetic permeability: The ratio of the magnetic induction, B , to the inducing field strength or magnetizing force, H ; permeability is dimensionless.

Phase Angle: The angle between the primary horizontal magnetic field and the horizontal electric field.

Resistivity: Property of a material which resists the flow of electrical current; units are ohm-m.

Skin depth: The depth at which the amplitude of a plane wave has been attenuated to $1/e$.

Wave impedance: Impedance at the surface of the Earth; differs from intrinsic impedance if the Earth is layered.

ATTACHMENT A: LISTING OF THE CODE FOR THE VLF.BAS PROGRAM

```
10 CLS
20 WIDTH 40
30 COLOR 15,0:LOCATE 10,9,0:PRINT CHR$(213)+STRING$(21,205)+CHR$(184)
40 LOCATE 11,9,0:PRINT CHR$(179)+"          VLF.BAS          "+CHR$(179)
50 LOCATE 12,9,0:PRINT CHR$(179)+STRING$(21,32)+CHR$(179)
60 LOCATE 13,9,0:PRINT CHR$(179)+"          Version 2.00          "+CHR$(179)
70 LOCATE 14,9,0:PRINT CHR$(212)+STRING$(21,205)+CHR$(190)
80 PRINT
90 PRINT
100 PRINT
110 PRINT "Please press Return to continue."
120 PRINT
130 PRINT
140 PRINT
150 INPUT P$
160 WIDTH 80
170 PRINT
180 PRINT
190 PRINT
200 PRINT "FORWARD MODELING COMPUTER PROGRAM FOR THE VERY LOW FREQUENCY"
210 PRINT "RADIO WAVE EARTH RESISTIVITY ELECTROMAGNETIC METHOD:"
    PRINT "VERSION 2.00"
220 PRINT
230 PRINT
240 PRINT "By Deborah G. Grantham, F. P. Haeni and David L. Mazzaferro,
    1986"
250 PRINT
260 PRINT "U. S. GEOLOGICAL SURVEY, HARTFORD, CONNECTICUT"
270 PRINT
280 PRINT
290 PRINT
300 PRINT "VLF.BAS calculates the apparent resistivity and quadrature"
310 PRINT "phase angle that would be measured by very low frequency"
320 PRINT "(VLF) Earth resistivity instruments at the surface of the"
330 PRINT "Earth for a given sequence of conductive or resistive"
340 PRINT "horizontal layers."
350 PRINT
355 PRINT "Acknowledgment is due Dr. Frank Frischknecht, U.S."
360 PRINT "Geological Survey, Denver, Colorado, for providing"
365 PRINT "us with the algorithm for VLF.BAS."
370 PRINT
380 PRINT "Please press Return to continue."
390 PRINT
420 INPUT P$
430 REM
440 CLS
```

```

450 WIDTH 80
460 PRINT
470 PRINT
480 PRINT
490 PRINT
500 PRINT
510 PRINT "A maximum of 10 layers and 20 radio frequencies can be"
520 PRINT "accommodated. Values for thickness are in meters, values"
530 PRINT "for resistivity are in ohm-meters and values for"
540 PRINT "frequency are in kiloHertz. The thickness of the"
550 PRINT "deepest layer is arbitrarily assigned a value of"
560 PRINT "10,000 meters."
570 PRINT
580 PRINT
590 PRINT
600 PRINT
610 PRINT
620 PRINT "ALL RESPONSES MUST BE IN UPPER CASES LETTERS."
630 PRINT
640 PRINT
650 PRINT
660 PRINT "Please press Return to continue."
670 INPUT P$
680 PRINT
690 PRINT
700 CLS
710 REM -----
720 REM DEFINITION OF VARIABLES
730 REM M: NUMBER OF LAYERS
740 REM F: NUMBER OF RADIO FREQUENCIES
750 REM P: ARRAY HOLDING VALUES FOR RESISTIVITY AND THICKNESS FOR EACH
760 REM Q: FREQUENCY
770 REM X: APPARENT RESISTIVITY
780 REM W: PHASE ANGLE
790 REM D1,D2: INVERSE OF SKIN DEPTHS
800 REM V: IMAGINARY PART OF WAVE IMPEDANCE
810 REM U: REAL PART OF WAVE IMPEDANCE
820 REM B, C, T1, T2, G, R, S, H, K, L, C1, A ARE INTERMEDIATE VALUES
830 REM USED IN CALCULATING THE APPARENT RESISTIVITY AND PHASE ANGLE
840 REM -----
850 Q$ = "S"
860 JK = 1
870 PRINT
880 PRINT
890 PRINT
900 PRINT "If you wish to use the printer, make sure it is ready.
          You may have to exit the program and key in 'Ctrl P'."
910 PRINT
920 PRINT

```

```

930 PRINT "Please enter 'P, in upper case, if you want printed
        output.  If you don't want printed output, press return."
940 PRINT
950 INPUT W$
960 PRINT
970 PRINT
980 CLS
990 PRINT "Enter the site name, 40 characters or fewer,
        no commas or semi-colons."

1000 PRINT
1010 PRINT
1020 INPUT SN$
1030 PRINT
1040 PRINT
1050 OPTION BASE 1
1060 PRINT "Enter the total number of layers
        assumed to be present.  Ten (10) is maximum."

1070 PRINT
1080 PRINT
1090 INPUT M$
1100 PRINT
1110 PRINT
1120 PRINT
1130 M = VAL(M$)
1140 IF M = 0 OR M > 10 GOTO 1160
1150 GOTO 1210
1160 PRINT "The value you have entered is unacceptable.
        Please press Return and try again."

1170 INPUT P$
1180 GOTO 1060
1190 PRINT
1200 PRINT
1210 PRINT "Enter the total number of radio
        frequencies used.  Twenty (20) is maximum."

1220 PRINT
1230 PRINT
1240 INPUT F$
1250 F = VAL(F$)
1260 IF F = 0 OR F > 20 GOTO 1280
1270 GOTO 1310
1280 PRINT "The value you have entered is
        unacceptable.  Please press Return and try again."

1290 INPUT P$
1300 GOTO 1210
1310 CLS
1320 DIM P(10,2),Q(20)
1330 CLS
1340 FOR I=1 TO M
1350 FOR J=1 TO 2

```

```

1360 IF I=M AND J=2 THEN GOTO 1450
1370 IF J=1 THEN J$ = "resistivity"
1380 IF J=1 THEN U$ = "ohm/meters"
1390 IF J=2 THEN J$ = "thickness"
1400 IF J=2 THEN U$ = "meters"
1410 PRINT "Enter the value for ";J$;"
      in layer number ";I;", in ";U$
1420 PRINT
1430 INPUT P(I,J)
1440 GOTO 1460
1450 P(M,2)=10000
1460 CLS
1470 NEXT J
1480 NEXT I
1490 IF Q$ = "S" GOTO 1510
1500 IF Q$ = "C" GOTO 1830
1510 FOR E=1 TO F
1520 PRINT "Enter the value for radio
      frequency number ";E;" in kilohertz."
1530 PRINT
1540 PRINT "COMMONLY USED RADIO FREQUENCIES ARE SHOWN BELOW."
1550 PRINT
1560 PRINT "  CUTLER, MAINE          17.8 KILOHERTZ  (BEFORE 1985)
      CUTLER, MAINE          24.0    do.      (AS OF 1985)
      ANNAPOLIS, MARYLAND    21.4    do.      "
1570 PRINT "  LUALUALEI, HAWAII      23.4    do."
1580 PRINT "  SEATTLE, WASHINGTON    18.6    do."
1590 PRINT
1600 INPUT Q(E)
1610 Q(E) = Q(E)*1000
1620 PRINT
1630 NEXT E
1640 PRINT "FREQUENCIES"
1650 FOR E = 1 TO F
1660 PRINT Q(E)/1000
1670 NEXT E
1680 E = 1
1690 PRINT
1700 PRINT
1710 PRINT
1720 PRINT "Are these frequencies correct?  Type Y/N and Return."
1730 INPUT YN$
1740 PRINT
1750 PRINT
1760 PRINT
1770 IF YN$ = "N" GOTO 1520
1780 PRINT
1790 PRINT
1800 PRINT

```



```

1820 PRINT
1830 CLS
1840 PRINT SN$
1850 IF W$ = "P" GOTO 1870
1860 GOTO 1880
1870 LPRINT SN$
1880 PRINT
1890 IF W$ = "P" GOTO 1910
1900 GOTO 1920
1910 LPRINT
1920 PRINT "Data Summary For Run Number ";JK
1930 IF W$ = "P" GOTO 1950
1940 GOTO 1960
1950 LPRINT "Data Summary For Run Number ";JK
1960 PRINT
1970 IF W$ = "P" GOTO 1990
1980 GOTO 2000
1990 LPRINT
2000 PRINT "FREQUENCY";TAB(15);"RESISTIVITY";TAB(28);"PHASE ANGLE"
2010 IF W$ = "P" GOTO 2030
2020 GOTO 2070
2030 LPRINT "FREQUENCY";TAB(20);"RESISTIVITY";TAB(38);"PHASE ANGLE"
2040 REM -----
2050 REM VLF.BAS ALGORITHM
2060 REM -----
2070 FOR N1=1 TO F
2080 R=1
2090 S=0
2100 FOR N2=0 TO M-2
2110 D1=SQR(39.4784*.0000001*Q(N1)/P(M-N2-1,1))
2120 D2=SQR(39.4784*.0000001*Q(N1)/P(M-N2,1))
2130 B=D1*P(M-N2-1,2)
2140 C=EXP(2*B)+EXP(-(2*B))+2*COS(2*B)
2150 T1=(EXP(2*B)-EXP(-(2*B)))/C
2160 T2=2*SIN(2*B)/C
2170 G=D1*(R-S)+D2*(T1-T2)
2180 H=D1*(R+S)+D2*(T1+T2)
2190 K=D2+D1*(T1*(R-S)-T2*(R+S))
2200 L=D2+D1*(T2*(R-S)+T1*(R+S))
2210 C1=K*K+L*L
2220 R=(G*K+H*L)/C1
2230 S=(H*K-G*L)/C1
2240 NEXT N2
2250 A=SQR(39.4748*.0000001*Q(N1)*P(1,1))
2260 U=A*(R-S)
2270 V=A*(R+S)
2280 REM -----
2290 REM PHASE ANGLE
2300 W=ATN(V/U)*57.2958

```

```

2310 REM -----
2320 REM APPARENT RESISTIVITY
2330 X=(U*U+V*V)/(78.9568*Q(N1)*.0000001)
2340 REM -----
2350 PRINT
2360 PRINT USING "##.##^^^"      ";Q(N1);X;W
2370 IF N1=5 GOTO 2420
2380 IF N1=10 GOTO 2420
2390 IF N1=15 GOTO 2420
2400 GOTO 2470
2410 CLS
2420 PRINT
2430 PRINT "Press return to continue."
2440 PRINT
2450 INPUT P$
2460 CLS
2470 IF W$ = "P" GOTO 2490
2480 GOTO 2500
2490 LPRINT USING "##.##^^^"      ";Q(N1);X;W
2500 NEXT N1
2510 PRINT
2520 PRINT "Press Return to continue."
2530 PRINT
2540 INPUT P$
2550 IF W$ = "P" GOTO 2570
2560 GOTO 2580
2570 LPRINT
2580 PRINT
2590 IF W$ = "P" GOTO 2610
2600 GOTO 2620
2610 LPRINT
2620 CLS
2630 PRINT " LAYER PARAMETERS "
2640 PRINT
2650 IF W$ = "P" GOTO 2670
2660 GOTO 2690
2670 LPRINT " LAYER PARAMETERS "
2680 LPRINT
2690 FOR I=1 TO M
2700 FOR J=1 TO 2
2710 PRINT P(I,J)
2720 IF I=5 AND J=2 GOTO 2740
2730 GOTO 2790
2740 PRINT
2750 PRINT "Press Return to continue."
2760 PRINT
2770 INPUT P$
2780 CLS
2790 IF W$ = "P" GOTO 2810

```

```

2800 GOTO 2820
2810 LPRINT P(I,J)
2820 NEXT J
2830 PRINT
2840 IF W$ = "P" GOTO 2860
2850 GOTO 2870
2860 LPRINT
2870 NEXT I
2880 PRINT
2890 JK = JK+1
2900 PRINT
2910 PRINT
2920 PRINT "Press Return to continue."
2930 PRINT
2940 PRINT
2950 INPUT P$
2960 CLS
2970 PRINT
2980 PRINT
2990 PRINT "Enter 'C' to Change one or more
        variables in the same problem"
3000 PRINT
3010 PRINT
3020 PRINT "Enter 'N' to input all New data"
3030 PRINT
3040 PRINT
3050 PRINT "Enter 'E' to Exit the program."
3060 PRINT
3070 INPUT Q$
3080 IF Q$="C" GOTO 3210
3090 IF Q$ = "N" GOTO 1330
3100 IF Q$ = "E" GOTO 3610
3110 PRINT
3120 CLS
3130 PRINT "Please enter 'C' or 'E',or
        'N' only; entries like ";Q$;" are invalid."
3140 PRINT
3150 PRINT "Press return and try again."
3160 INPUT P$
3170 CLS
3180 GOTO 2990
3190 CLS
3200 END
3210 PRINT "PRESENT PARAMETER VALUES"
3220 PRINT
3230 FOR I=1 TO M
3240 FOR J=1 TO 2
3250 PRINT P(I,J)
3260 NEXT J

```

```

3270 PRINT
3280 NEXT I
3290 PRINT "Are these values OK? (Y/N)"
3300 INPUT YN$
3310 IF YN$="Y" GOTO 1820
3320 PRINT "You will correct one parameter at a time"
3330 PRINT "Type the number of the layer
           for which a change will be made"
3340 INPUT X$
3350 X = VAL(X$)
3360 IF X = 0 GOTO 3400
3370 FOR I=1 TO M
3380 IF I=X GOTO 3430
3390 NEXT I
3400 PRINT "The value you have entered is
           not acceptable. Press return and try again."
3410 INPUT P$
3420 GOTO 3320
3430 IF X<M GOTO 3450
3440 IF X=M GOTO 3470
3450 PRINT "Enter T to change layer
           ";X;" thickness, or R to change its resistivity"
3460 GOTO 3480
3470 PRINT "Enter R to change its resistivity."
3480 INPUT TR$
3490 IF TR$="R" GOTO 3530
3500 IF TR$="T" GOTO 3560
3510 PRINT "Please enter 'T' or 'R' only; entries
           like ";TR$;" are invalid"
3520 GOTO 3480
3530 PRINT "New Resistivity?"
3540 INPUT P(X,1)
3550 GOTO 3210
3560 PRINT "New Thickness?"
3570 INPUT P(X,2)
3580 GOTO 3210
3590 CLS
3600 GOTO 3020
3610 CLS
3620 END

```